SPIM supports a subset of the MIPS assembler directives:

.align n 
Align the next datum on a 2n byte boundary. For example, .align 2 aligns the next value on a word boundary. .align 0 turns off automatic alignment of .half, .word, .float, and .double directives until the next .data or

.kdata directive.

.ascii str 
Store the string str in memory, but do not null-terminate it.

.asciiz str 
Store the string str in memory and null-terminate it.

.byte b1,..., bn Store the n values in successive bytes of memory.

.data <addr> Subsequent items are stored in the data segment. If the optional argument addr is present, subsequent items are stored starting at address addr.

.double d1, ..., dn Store the n floating-point double precision numbers in successive memory locations.

.extern sym size Declare that the datum stored at sym is size bytes large and is a global label. This directive enables the assembler to store the datum in a portion of the data segment that is efficiently accessed via register $gp.

.float f1,..., fn Store the n floating-point single precision numbers in successive memory locations. 

.globl sym Declare that label sym is global and can be referenced from other files.  

half h1, ..., hn Store the n 16-bit quantities in successive memory halfwords.

.kdata <addr> Subsequent data items are stored in the kernel data segment. If the optional argument addr is present, subsequent items are stored starting at address addr.

.ktext <addr> Subsequent items are put in the kernel text segment. In SPIM, these items may only be instructions or words (see the .word directive below). If the optional argument addr is present, subsequent items are stored starting at address addr. 

.set noat and .set at The first directive prevents SPIM from complaining about subsequent instructions that use register $at. The second directive reenables the

warning. Since pseudoinstructions expand into code that uses register $at, programmers must be very careful about leaving values in this register. 

.space n Allocate n bytes of space in the current segment (which must be the data segment in SPIM).

.text <addr> Subsequent items are put in the user text segment. In SPIM, these items may only be instructions or words (see the .word directive below). If the optional argument addr is present, subsequent items are stored starting at address addr.

.word w1,..., wn Store the n 32-bit quantities in successive memory words.

Spim example - conio.asm

# conio - console input and output

        .data

a0wrds: .word   1,2,3,0     #Words pointed to by $a0

tst:    .asciiz "Input an integer. \n"

done:   .asciiz "\n Program Terminated \n"

chrmsg: .asciiz "Input a character to be converted to upper case or press Esc key.\n"

TEMP:   .space  256         #Reserve 256 bytes

        .text               #Code goes in text segment

        .globl  main        

##################################################

main:

        sw      $ra,0($sp)  #Push return address

        addiu   $sp,$sp,-4  # onto stack

#

loop:   la      $a0,chrmsg  #Point $a0 to message

        jal     wr_str      #Output request for a character

        jal     rd_chr      #Get character

        jal     crlf        #Go to new line

        addi    $t0,$0,0x1b #Put escape char in $t0

        beq     $a0,$t0,doint #Go to next if escape

        jal     UPPER       #Convert character to upper case

        jal     WR_CHR      #Output upper case character

        jal     crlf        #Go to new line

        j       loop

#

doint:  la      $a0,tst     #Use pseudoinstruction

        jal     wr_str      #OUTPUT TEST STRING TO CONSOLE

        jal     RD_INT      #Read integer

        jal     crlf

        jal     WR_INT      #Output integer

        la      $a0,done    #Point $a0 to done message

        jal     wr_str      #Output done message

#

        addiu   $sp,$sp,4   #Pop return address

        lw      $ra,0($sp)  # from stack

        jr      $ra         #Return              

###############################################################

# 

# FUNCTION: wr_str - OUTPUT 0 TERMINATED STRING TO CONSOLE

#    INPUT: $A0 - POINTER TO STRING

#   OUTPUT: NONE

#    CALLS: SYSCALL

# DESTROYS: $V0

#

###############################################################

wr_str:     

        addi    $v0,$0,4    #PRINT STRING REQUEST CODE

        syscall

        jr      $ra         #Return

###############################################################

# 

# FUNCTION: crlf - OUTPUT 0 TERMINATED STRING TO CONSOLE

#    INPUT: NONE

#   OUTPUT: NONE

#    CALLS: wr_str

# DESTROYS: $A0, $V0

#

###############################################################

        .data

crlfz:  .ascii  "\n" 

        .text       

crlf:

        sw      $ra,0($sp)  #Push return address

        addiu   $sp,$sp,-4  # onto stack

        sw      $a0,0($sp)  #Push $a0 

        addiu   $sp,$sp,-4  # onto stack

        la      $a0,crlfz   #Point to new line string

        jal     wr_str      #Output CRLF to console

        addiu   $sp,$sp,4   #Pop $a0

        lw      $a0,0($sp)  # from stack

        addiu   $sp,$sp,4   #Pop return address

        lw      $ra,0($sp)  # from stack

        jr      $ra         #Return              

###############################################################

# 

# FUNCTION: RD_INT - READ INTEGER FROM KEYBOARD

#    INPUT: NONE

#   OUTPUT: $a0 - integer read from console

#    CALLS: SYSCALL

# DESTROYS: $v0

#

###############################################################

RD_INT:     

        addi    $v0,$0,5    #Read integer request code

        syscall

        add     $a0,$v0,$0  #Move integer into $a0

        jr      $ra         #Return

###############################################################

# 

# FUNCTION: WR_INT - Otput integer to console

#    INPUT: $a0 - integer

#   OUTPUT: NONE

#    CALLS: SYSCALL

# DESTROYS: $v0

#

###############################################################

WR_INT:     

        addi    $v0,$0,1    #Print integer request code

        syscall

        jr      $ra         #Return

###############################################################

# FUNCTION: rd_chr - READ Character from KEYBOARD

#    INPUT: NONE

#   OUTPUT: $a0 - Character read from console

#    CALLS: SYSCALL

# DESTROYS: $v0

###############################################################

rd_chr:     

        addi    $v0,$0,12   #Read char request code

        syscall

        add     $a0,$v0,$0  #Move character in $v0 to $a0

        jr      $ra         #Return

###############################################################

# FUNCTION: WR_CHR - Output character to console

#    INPUT: NONE

#   OUTPUT: $a0 - Character to output to console

#    CALLS: SYSCALL

# DESTROYS: $v0

###############################################################

WR_CHR:     

        addi    $v0,$0,11   #Print character request code

        syscall

        jr      $ra         #Return

###############################################################

# FUNCTION: WR_HEX - Otput integer in Hex to console

#    INPUT: $a0 - integer

#   OUTPUT: NONE

#    CALLS: SYSCALL

# DESTROYS: $v0

###############################################################

WR_HEX: 

        addi    $v0,$0,1    #Print integer request code

        syscall

        jr      $ra         #Return

###########################################################

# FUNCTION: UPPER - IF ASCII character (see Fig 3.15) in $a0

#           is between 'a' and 'z' convert to upper case

#           else $a0 is unchanged.

#           Note: A lower case ASCII character can be 

#           converted to upeercase by subtracting 32.

#   INPUTS: $a0 - Contains 8 bit ASCII character in least

#           significant byte. High order bytes are 0.

#  OUTPUTS: $a0 - Contains input ASCII character converted

#           to upper case.

# DESTROYS: $a0, $s0

###########################################################

#

UPPER:

        slti    $s0,$a0,'a' #Compare input char with 'a'

        bgtz    $s0,Done    #Exit if less than 'a'

        slti    $s0,$a0,'z'+1 #Compare input char with 'z'

        blez    $s0,Done    #Exit if greater than 'z'

        addi    $a0,-32     #Convert char to upper case.

Done:   jr      $ra         #Return

        .data

